NAME: _____ STUDENT #: _____

EECE 259: Introduction to Microcomputers **Lecture Quiz** **Feb 4, 2011**

1. Write a program that computes **(3*A-B*C)/D** and puts the answer in register r2. The symbols have been defined below. Your program should also work for other reasonable values of **A, B, C,** or **D**.

```
.equ    A, 7
.equ    D, 5
.global _start
.text
_start:         movi    r2, 3
                muli    r2, r2, A               /* 3*A */

                movia   r3, B
                ldw     r3, 0(r3)               /* retrieve B=2 */

                movia   r4, C
                ldw     r4, 0(r4)               /* retrieve C=3 */

                mul     r3, r3, r4              /* B*C */
                sub     r2, r2, r3              /* 3*A – B*C */

                movi    r3, D
                div     r2, r2, r3              /* no divi instruction */

STOP:           br      STOP

.data
B:
.word   2
C:
.word   3
.end
```

2. Write a short program to compute N!, e.g. 5! = 1*2*3*4*5. Store the answer in memory at label FACT:

```
.equ    N, 5
.global _start
.text
_start:         movi    r4, 1           /* loop iteration counter */
                movi    r5, N           /* loop bound */
                movi    r6, 1           /* current factorial product */

loop:           addi    r4, r4, 1
                mul     r6, r6, r4
                blt     r4, r5, loop    /* note: blt is better than bne */

                movia   r7, FACT
                stw     r6, 0(r7)

STOP:           br      STOP

.data
FACT:
.word   0       /* alternatively, use ".skip 4", where 4 = # of bytes */
.end

/* Note: the code above has bugs: it doesn't work for N=0 or N=1. */
/* How would you change the program to fix it? */
```

Give 15 minutes (closed book) to write this quiz.
Then give 15 minutes (open book) to discuss with friend/neighbor.
Taking up each solution requires 10 minutes.